Module 12 Taylor-ontwikkelingen

Onderwerp	Taylor-ontwikkelingen van functies van één of meer variabelen.
Voorkennis	Taylor-ontwikkelingen.
Expressies	`taylor`, `convert(expressie,polynom)`, `mtaylor`, `coefftayl`
Zie ook	Module 7, 23.

12.1 Taylor-ontwikkelingen

Zij $f : I \subset \mathbb{R} \to \mathbb{R}$ een $n+1$ maal continu differentieerbare functie met continue afgeleiden. Dan geldt voor elke $a \in I$ dat

$$f(a+h) = \sum_{k=0}^{n} \frac{h^k}{k!} f^{(k)}(a) + o(h^n), \quad (h \to 0).$$

Deze som noemen we de n^{de} orde Taylor-ontwikkeling van f rond het punt $x = a$. De stelling is hier geformuleerd met het ‘kleine ordesymbool’, dat betekent dat

$$\lim_{h \to 0} \frac{f(a+h) - \sum_{k=0}^{n} \frac{h^k}{k!} f^{(k)}(a)}{h^n} = 0.$$

Dit wordt ook wel met het ‘grote ordesymbool’ genoteerd:

$$f(a+h) = \sum_{k=0}^{n} \frac{h^k}{k!} f^{(k)}(a) + \mathcal{O}(h^{n+1}), \quad (h \to 0).$$

12.2 Taylor-ontwikkeling in Maple

De n^{de} orde Taylor-ontwikkeling van een functie f rond een punt $x = a$ wordt verkregen via

`taylor`

```
taylor( f(x), x=a, n+1 )
```

Maple werkt met het grote ordesymbool; we moeten $n+1$ opgeven om de n^{de} orde Taylor-ontwikkeling (dus met $\mathcal{O}(h^{n+1})$) te krijgen. Vaak zullen we dit resultaat zonder ordeterm willen hebben, bijvoorbeeld om een plaatje te kunnen tekenen. De Taylor-ontwikkeling zonder ordeterm is gewoon een polynoom. We verkrijgen deze polynoom met `convert(expressie,polynom)`. Door middel van `unapply` is er weer een functie van te maken.

`convert`, `polynom`

Voorbeeldopgave

Definieer een functie g die de 10^{de} orde Taylor-ontwikkeling is van e^x rond het punt $x = 0$. Bereken $g(2)$.

Voorbeeldsessie

```
> hulp := taylor(exp(x), x=0, 11);
```

$$hulp := 1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 + \frac{1}{720}x^6 + \frac{1}{5040}x^7 + \frac{1}{40320}x^8 + \frac{1}{362880}x^9 + \frac{1}{3628800}x^{10} + \mathrm{O}(x^{11})$$

```
> convert(hulp, polynom); g := unapply(%,x);
```

$$g := x \to 1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 + \frac{1}{720}x^6 + \frac{1}{5040}x^7 + \frac{1}{40320}x^8 + \frac{1}{362880}x^9 + \frac{1}{3628800}x^{10}$$

```
> g(2);
```

$$\frac{34913}{4725}$$

Toelichting

Het resultaat van het commando `convert` is een uitdrukking waar `x` in voorkomt. De procedure `unapply` maakt hiervan de gevraagde functie van `x`. ◇

12.3 Functies van meer variabelen

`mtaylor`

Het is ook mogelijk om Taylor-ontwikkelingen te berekenen van functies van meer dan één variabele. Dit moet met het commando `mtaylor`. De aanroep is:

```
mtaylor( f(x1, ..., xn), [x1=a1, ..., xn=an], orde)
```

Hierbij zijn $a_1, \ldots, a_n$ de coördinaten van het punt rond welk de Taylor-ontwikkeling gemaakt wordt. De parameter `orde` specificeert de orde van de Taylor-ontwikkeling. In tegenstelling tot het ééndimensionale geval bevat het resultaat van `mtaylor` géén ordeterm en is gewoon een polynoom. We geven een voorbeeld.

Voorbeeldsessie

```
> f := (x,y,z) -> (1 + x + y + z)^6;
```

$$f := (x,\, y,\, z) \to (1 + x + y + z)^6$$

```
> taylor_ontw := mtaylor( f(x,y,z), [x=0, y=0, z=0], 3);
```

$$taylor\_ontw := 1 + 6\,x + 6\,y + 6\,z + 15\,x^2 + 30\,y\,x + 30\,z\,x + 15\,y^2 + 30\,z\,y + 15\,z^2$$

```
> f_t := unapply(taylor_ontw, x,y,z):
  f( 0.1, 0.1, 0.1), f_t( 0.1, 0.1, 0.1);
```

$$4.826809,\ 4.15$$

De coëfficiënt van $x\,y^2\,z$:

```
> coeftayl( f(x,y,z), [x,y,z]=[0,0,0], [1,2,1] );
```

$$180$$

Zonder gebruik te maken van `coefftayl`, dus op de manier van §11.4:

```
> u := f(x,y,z):
  coeff(u,x,1);
```

$$2\,(1 + y + z)^3\,((1 + y + z)\,(2 + 2\,y + 2\,z) + (1 + y + z)^2)$$

```
> coeff(coeff(u,x,1),y,2);
```

$$6\,(1 + z)^3 + 2\,((1 + z)\,(2 + 2\,z) + (1 + z)^2)\,(6 + 6\,z) + 2\,(3 + 3\,z)\,((1 + z)\,(2 + 2\,z) + (1 + z)^2)$$

```
> coeff(coeff(coeff(u,x,1),y,2),z,1);
```

$$180$$

Toelichting

De *derde* orde (grote $\mathcal{O}$!) Taylor-ontwikkeling geeft dus de termen waarvan de som van de machten van x, y en z niet groter dan 2 is.

`coeftayl`

De procedure `coeftayl` kan worden gebruikt om de coëfficiënt van $x^{\alpha_1} y^{\alpha_2} z^{\alpha_3}$ in de Taylor-ontwikkeling van $f(x, y, z)$ te berekenen. Het commando

```
coeftayl( u, [x1,..., xn]=[a1,..., an], [c1,..., cn] )
```

geeft de coëfficiënt van $(x_1 - a_1)^{c_1} \ldots (x_n - a_n)^{c_n}$ van de Taylor-ontwikkeling van de expressie `u`, opgevat als functie van $\mathbf{x}$, rond het punt $\mathbf{x} = \mathbf{a}$. Ook als de expressie `u` een polynoom is, dan vereist deze methode minder tikwerk dan wanneer we `coeff` (zie §11.4) zouden gebruiken. ◇

Opgave 12.1

Gegeven de functie $f : x \mapsto \log(1 + \tan(x))$.

(a) Bepaal de Taylor-ontwikkeling van f rond $x = 0$ tot en met de term met x^4.
(b) Zelfde vraag, maar nu rond $x = 1/2$.

Opgave 12.2

Teken in één figuur de grafieken van $\sin(x)$, en de Taylor-ontwikkelingen van $\sin(x)$ rond $x = 1$ van de orde 1 en 5. (Dus tot en met de term met $x - 1$ respectievelijk $(x - 1)^5$.) Neem in de figuur voor x het interval $[0, 2\pi]$ en voor y het interval $[-2, 2]$.

Opgave 12.3

Teken in één figuur de grafieken van $f : x \mapsto e^{-1/x^2}$, en de Taylor-ontwikkelingen van f rond $x = \frac{1}{10}$ van de orde 3 en 7. (Dus tot en met de term met $(x - \frac{1}{10})^3$ respectievelijk met $(x - \frac{1}{10})^7$.) Neem in de figuur voor x het interval $[-2, 2]$. Zorg dat de drie grafieken verschillende kleuren krijgen. (*Aanwijzing:* kies `axes=boxed` zodat u goed kunt zien hoe de grafieken er in de buurt van de x-as uitzien.) Teken de grafieken van de Taylor-polynomen ook apart. Zijn deze polynomen monotoon stijgend of monotoon dalend? Verklaar het resultaat! (Let op de schaalverdeling langs de y-as.)
Aanwijzing: Bereken de 5^{de} en de 10^{de} afgeleide van f in het punt $x = 0$. Formuleer een vermoeden en bewijs het.

Opgave 12.4

Bepaal de Taylor-ontwikkeling van $\frac{e^x-1}{x}$ rond $x = 0$ tot en met de term met x^2. Wat kunt u concluderen over

$$\lim_{x\to 0} \frac{e^x - 1}{x}?$$

Opmerking: Het zal u opvallen dat de (grote) orde 4 opgegeven moet worden om de Taylor-ontwikkeling tot en met de term met x^2 te krijgen. Dat komt omdat Maple eerst de Taylor-ontwikkeling van de teller uitrekent en vervolgens elke term door x deelt.

Opgave 12.5

Gegeven de functie $f : x \mapsto \sin(x)$.

(a) Bepaal de Taylor-polynoom van f rond $x = a \in [-1, 2]$ tot en met de term met $(x - a)^3$. Noem het resultaat g.

(b) Ga na dat

$$\epsilon(g) := \int_{-1}^{2} (f(x) - g(x))^2 \, dx$$

een goed criterium is om de functies $f(x)$ en $g(x)$ op het interval $[-1, 2]$ met elkaar te vergelijken. Waarom is

$$\int_{-1}^{2} f(x) - g(x) \, dx$$

geen goed criterium?

(c) Voor welke a is $\epsilon(g)$ het kleinst?
Aanwijzing: Teken de grafiek van $\epsilon(g)$ als functie van a. Verander het interval, en desnoods ook de range in verticale richting, zodanig dat u kunt zien waar het globale minimum ligt. Vind een numerieke benadering voor dit minimum (gebruik `fsolve`).

Opgave 12.6

Bepaal de derde orde Taylor-ontwikkeling van de volgende functies:

(a) $f(x, y) = \sqrt{1 + x^2 + \sin(y)}$, rond $(0, 0)$,

(b) $f(x, y, z) = xe^y \sin(z)$, rond $(1, 0, 0)$.